

Modul Praktikum



PEMROGRAMAN VISUAL MENGGUNAKAN MICROSOFT VISUAL BASIC 6

**PRODI TEKNIK SIPIL
FAKULTAS TEKNIK
UNIVERSITAS SEBELAS MARET**

Pengantar


Pengantar pada Lingkungan Visual Basic 6

Tujuan Praktikum

- Mempelajari dasar-dasar desain dan penerapan Visual Basic 6
- Mempelajari penggunaan toolbox.
- Mempelajari cara memodifikasi object properties.
- Mempelajari object method.
- Menggunakan menu design window.

Struktur Program di Visual Basic 6

Project (.VBP, .MAK)



Application (Project) terdiri atas:

- **Form** – Window dimana programmer membuat user interface
- **Control** – bentuk Grafis yang digambarkan pada form untuk membuat interaksi dengan user (text boxes, labels, scroll bars, command buttons, dsb.) (Form dan Control adalah **object**)
- **Properties** – Merupakan karakteristik yang dimiliki oleh form/kontrol. Contoh property termasuk names, captions, size, color, dan position.
- **Method** – Merupakan procedure tersedia di Visual Basic 6 yang dapat dipanggil untuk mengerjakan suatu tindakan pada object tertentu.
- **Event Procedure** - Code yang berkaitan dengan object, Code ini akan dieksekusi ketika suatu event terjadi.
- **General Procedures** - Code yang tidak berkaitan secara langsung dengan object.
- **Modules** – Merupakan kumpulan dari general procedure.

Langkah dalam Membangun Aplikasi


Ada tiga langkah dalam membangun aplikasi dalam Visual Basic 6:

1. **Mendesain interface**
2. **Mengatur properties** control
3. **Menuliskan code** pada control


Menggambar User Interface dan Mengatur Property

- Visual Basic 6 berjalan dalam tiga mode.


- ⇒ **Design** mode – untuk membangun aplikasi
 - ⇒ **Run** mode – untuk menjalankan aplikasi
 - ⇒ **Break** mode – aplikasi berhenti sementara dan debugger muncul
- **Form Window** merupakan pusat untuk membangun aplikasi Visual Basic. Tempat dimana kita menggambar aplikasi kita.




- **Toolbox** adalah menu yang berisi kontrol-kontrol yang digunakan di dalam aplikasi kita.




- **Properties Window** digunakan untuk menempatkan default properties untuk object. drop-down box pada bagian atas window berisi list dari semua object yang ada pada form. Ada dua tampilan: **Alphabetic** (urutan abjad) and **Categorized** (urutan kelompok). Pada bagian bawah window berisi properties untuk object terkait.




- **Form Layout Window** menunjukkan (saat eksekusi program) form kita akan ditampilkan pada layar monitor:



- **Project Window** menampilkan list dari semua form dan module yang dimiliki oleh suatu project. Kita juga bisa menampilkan pilihan **Form** atau **Code** window dari Project window.




- seperti telah disebutkan di muka, ada dua cara untuk menempatkan control pada form:
 1. Double-click salah satu tool pada toolbox dan akan muncul control tersebut pada form dengan ukuran default-nya. Kemudian kita dapat mengatur penampilannya di form.
 2. Click pada salah satu tool di toolbox, kemudian gerakkan mouse pointer ke bagian form window. Cursor akan berubah menjadi tanda +. Tempatkan cursor pada bagian atas-kiri di mana control tersebut akan ditempatkan. Press mouse sebelah kiri dan jangan dilepaskan saat menyeret cursor ke arah kanan-bawah. Saat kita melepaskan mouse maka control tersebut sudah tergambar di form.
- Untuk memindahkan posisi control yang ada di form, click object dan seret ke posisi yang baru kemudian lepaskan mouse.
- Untuk mengubah ukuran dari suatu control, click object sehingga ter-select dan tanda pengubah ukuran muncul. Gunakan tanda pengubah untuk mengubah ukuran object.



Mengatur Properties dari Object pada Waktu Desain

Properties dari object dapat ditentukan pada saat desain dengan mengubah properties yang dikehendaki dari list properties. Ada dua tampilan: **Alphabetic** (urutan abjad) dan **Categorized** (urutan kelompok). Pada bagian bawah window berisi list properties untuk object terkait.



Property yang cukup penting dari sebuah object adalah **name** (nama). Suatu cara untuk memudahkan dalam mengenali object adalah dengan menempatkan tiga prefix pada jenis object kemudian diikuti oleh nama yang kita tentukan. Panjang nama object bisa sampai dengan 40 karakter.

Object	Prefix	Contoh
Form	frm	frmWatch
Command Button	cmd, btn	cmdExit, btnStart
Label	lbl	lblStart, lblEnd
Text Box	txt	txtTime, txtName
Menu	mnu	mnuExit, mnuSave
Check box	chk	chkChoice
Combo box	cbo	cboInterest
List box	lst	lstCity

Setting Properties saat Run Time

– Kita juga bisa men-setting properties saat aplikasi sedang dalam keadaan **running**. Untuk melakukan ini, kita harus menuliskan kode, dengan format seperti berikut ini:

```
ObjectName.Property = NewValue
```

Contoh: misal kita akan mengganti property **BackColor** dari suatu form bernama **frmStart** dengan warna biru, kita ketikkan:

```
frmStart.BackColor = &HFF0000
```

Bagaimana Nama Variable (Name) dipakai pada Object Event

- Nama yang kita berikan pada suatu object digunakan oleh Visual Basic untuk mengatur suatu framework dari event-driven procedure (procedure yang tergantung dari event) dimana kita bisa menempatkan code program. Format untuk penulisan dari subroutine (semua object di Visual Basic adalah subroutine) adalah:

```
Sub ObjectName_Event (Optional Arguments)
    .
    .
End Sub
```

- Visual Basic menyediakan baris Sub dengan argument (jika ada) dan End Sub. Dimana di antara keduanya kita bisa menyisipkan kode.

Variable

Setelah object ditambahkan pada form, Visual Basic secara otomatis membuat kerangka untuk semua event procedure. Kita dengan mudah bisa menambahkan code pada event procedure dimana aplikasi kita beraksi terhadapnya. Dalam hal ini kita memerlukan **variable**.

- Variable yang dipakai dalam Visual Basic:
 - ⇒ Tidak lebih dari 40 huruf
 - ⇒ Bagian pertama harus huruf
 - ⇒ Selanjutnya bisa terdiri atas huruf, angka, dan underscore (_)
 - ⇒ Tidak boleh memakai reserved word (kata yang diperlukan oleh Visual Basic)

Visual Basic Data Type

Data Type	Suffix	Keterangan
Boolean	None	TRUE , FALSE
Integer	%	Bilangan bulat
Long (Integer)	&	Bilangan bulat
Single (Floating)	!	Bilangan pecahan
Double (Floating)	#	Bilangan pecahan
Currency	@	Bilangan Mata uang
Date	None	Tanggal
Object	None	-
String	\$	Kata, Kalimat
Variant	None	-

Deklarasi Variable

- Ada tiga cara dalam mendeklarasikan variable:
 1. Default
 2. Implicit
 3. Explicit
- Jika variable tidak secara explicit atau implicit, maka secara default variable tadi dimasukkan tipe **variant**. Data tipe variant merupakan tipe khusus yang digunakan oleh Visual Basic yang dapat berisi numeric, string, atau date.
- Secara **implicit** bisa menggunakan tabel di atas. Contoh:

```
TextValue$ = "This is a string" ` membuat variable string
```

```
Amount% = 300 ` membuat variable integer
```

- Cara yang paling baik adalah mendeklarasikan variable secara **explicit**.
- Untuk mendeklarasikan variable secara **explicit**, pertama harus ditentukan **scope**-nya. Ada empat level scope:
 - ⇒ level Procedure
 - ⇒ level Procedure, Static
 - ⇒ level Form dan Module
 - ⇒ level Global
- Di dalam procedure, variable dideklarasikan menggunakan **Dim**:

```
Dim MyInt as Integer  
Dim MyDouble as Double  
Dim MyString, YourString as String
```

Saat procedure dimatikan maka semua nilai yang disimpannya akan hilang.

- Untuk membuat variable dengan procedure level, dan tetap mempertahankan nilai yang dibawanya ganti Dim dengan **Static**:

```
Static MyInt as Integer  
Static MyDouble as Double
```


- Untuk level Form (module), deklarasikan variable pada bagian **general** (bagian paling atas setelah keyword Option Explicit, dan gunakan **Dim**:

```
Dim MyInt as Integer  
Dim MyDate as Date
```

- Untuk level Globalvariable tetap menyimpan nilainya dan tersedia untuk semua procedure dalam aplikasi. Deklarasikan variable pada bagian **general** (bagian paling atas setelah keyword **Option Explicit**, dan gunakan keyword **Global**:

```
Global MyInt as Integer  
Global MyDate as Date
```


- Contoh scope dari Variable:



Procedure Routine1 punya akses ke X, Y, dan A (nilai hilang saat berhenti)
 Procedure Routine2 punya akses ke X, Y, dan B (nilai tetap)
 Procedure Routine3 punya akses ke X, Z, dan C (nilai hilang)

Pernyataan dan Ekspresi Visual Basic

Pernyataan yang paling sederhana disebut dengan **assignment**. Yaitu memberikan nilai pada suatu variable secara langsung. Perhatikan contoh di bawah ini.

```

StartTime = Now
Explorer.Caption = "Mencari File Tersembunyi"
BitCount = ByteCount * 8
Energy = Mass * LIGHTSPEED ^ 2
NetWorth = Assets - Liabilities

```

Pernyataan assignment ini menyimpan informasi/data pada variable terkait.

- Pernyataan dalam Visual Basic biasanya ditulis per-baris, tetapi dapat juga dipisahkan dengan menggunakan tanda colon (:). Contoh:

```

StartTime = Now : EndTime = StartTime + 10

```

- Jika pernyataan agak panjang, dapat dilanjutkan pada baris berikutnya dengan menggunakan alat penyambung spasi lalu diikuti dengan underscore (_). Contoh:

```

Months = Log(Final * IntRate / Deposit + 1) _
/ Log(1 + IntRate)

```

- Pernyataan berupa komentar/catatan bisa diawali dengan keyword **Rem** atau single quote ('). Contoh:

```

Rem Ini adalah komentar

```

```
' Ini Juga komentar
x = 2 * y ' cara lain menuliskan komentar
```

Function pada Visual Basic

Visual Basic mempunyai berbagai built-in **functions**. Beberapa contoh antara lain:

Function	Hasil	Contoh
Abs	Nilai absolut dari sebuah angka	Abs(-1) → 1
Asc	No kode ASCII / ANSI dari karakter	Asc("A") → 67
Chr	Karakter sesuai kode ASCII / ANSI	chr(65) → "A"
Cos	Nilai cos dari sudut dalam radian	Cos(3.14) → -1
Date	Tanggal saat ini dalam bentuk string	Date → 3/17/2007
Format	Date atau angka dikonversi ke string	Format(Date, "dd/mm/yyyy")
Left	Bagian kiri dari string	Left("Nama Saya", 4) → "Nama"
Len	Banyaknya huruf dalam string	Len("Nama") → 4
Mid	Sebagian dari string	Mid("Nama", 2, 2) → "am"
Now	Tanggal dan Waktu sekarang	Now → "3/17/2007 13:52:20"
Right	Bagian kanan dari string	Right("Nama Saya", 4) → "Saya"
Rnd	Angka acak	Int(101 * Rnd) + 100 → 171 'angka acak dari 100 s.d 200
Sin	Nilai sin dari sudut dalam radian	sin(3.14) → 0
Sqr	Akar dari angka	Sqr(4) → 2
Str	Angka dikonversi ke string	Str(10) → "10"
Time	Waktu sekarang sebagai string	Time → "14:01:02"
Timer	Jumlah detik yang berlangsung sejak 00.00	Timer → 50543.38
Val	Nilai dari string ke dalam angka	Val("Saya") → 0 Val("100") → 100

Percabangan pada Visual Basic – Pernyataan If

Pernyataan percabangan digunakan untuk melakukan sesuatu tindakan apabila suatu syarat terpenuhi.

– Cara yang paling sederhana pernyataan **If/Then**:

```
If TeganganIjin < Tegangan Then Print "Profil baja kurang"
```

Di sini, jika tegangan yang terjadi pada profil melampaui tegangan ijin maka pernyataan "Profil baja kurang" akan dicetak dilayar.

– Dapat Juga dengan menggunakan blok **If/Then/End If** untuk membuat banyak pernyataan:

```
If TeganganIjin < Tegangan Then
  Print "Profil baja kurang"
  Print "Coba ubah luas profil"
End if
```

Di sini, jika tegangan yang terjadi pada profil melampaui tegangan ijin maka dua pernyataan yang ada akan dicetak di layar.

- Atau gunakan blok **If/Then/Else/End If** :

```
If TeganganIjin < Tegangan Then
    Print "Profil baja kurang"
    Print "Coba ubah luas profil"
else
    Print "Profil baja memenuhi syarat"
End if
```

Di sini, jika tegangan yang terjadi pada profil melampaui tegangan ijin maka dua pernyataan yang ada akan dicetak di layar, tetapi jika berlaku sebaliknya maka "Profil baja memenuhi syarat" akan dicetak.

- Atau kita bisa menambahkan pernyataan **Elseif** :

```
If TeganganIjin < Tegangan Then
    Print "Profil baja kurang"
    Print "Coba ubah luas profil"
Elseif TeganganIjin = Tegangan then
    Print "Profil baja pas-pasan"
Else
    Print "Profil baja memenuhi syarat"
End if
```

Key Trapping

Key trapping dilakukan di dalam procedure **KeyPress** pada sebuah object. Misalkan suatu procedure pada form (suatu textbox **txtText**):

```
Sub txtText_KeyPress (KeyAscii as Integer)
    .
    .
    .
End Sub
```

Sebagai contoh, kita mempunyai textbox **txtExample** dan kita ingin bisa mengetikkan huruf kapital (kode ASCII adalah 65 s.d 90, atau dalam Visual Basic terdapat konstanta **vbKeyA** s.d **vbKeyZ**). Procedure keypress akan tampak seperti (suara **Beep** akan muncul apabila salah tekan tombol):

```
Sub txtExample_KeyPress(KeyAscii as Integer)
    If KeyAscii >= vbKeyA And KeyAscii <= vbKeyZ Then
        Exit Sub
    Else
        KeyAscii = 0
        Beep
    End If
End Sub
```

Di dalam key trapping, disarankan untk selalu membiarkan tombol backspace (ASCII code 8; Visual Basic **vbKeyBack**) untuk tidak dikunci, sehingga saat mengedit teks tidak mengalami masalah.

Select Case – Cara Lain untuk Percabangan

- Selain penggunaan If/Then/Else, kita juga bisa menggunakan format **Select Case** apabila banyak pilihan yang harus dilakukan.

- Misalkan kita telah menuliskan code menggunakan pernyataan **If**:

```
If Age = 5 Then
  Category = "Five Year Old"
ElseIf Age >= 13 and Age <= 19 Then
  Category = "Teenager"
ElseIf (Age >= 20 and Age <= 35) Or Age = 50 Or (Age >= 60 and Age
<= 65) Then
  Category = "Special Adult"
ElseIf Age > 65 Then
  Category = "Senior Citizen"
Else
  Category = "Everyone Else"
End If
```

Maka code yang bermakna sama dengan format **Select Case** adalah:


```
Select Case Age
  Case 5
  Category = "Five Year Old"
  Case 13 To 19
  Category = "Teenager"
  Case 20 To 35, 50, 60 To 65
  Category = "Special Adult"
  Case Is > 65
  Category = "Senior Citizen"
  Case Else
  Category = "Everyone Else"
End Select
```

Pernyataan GoTo

Pernyataan percabangan yang lain, yang tidak begitu populer adalah **GoTo**. Akan tetapi, hal ini diperlukan saat melakukan Run-Time error trapping. Formatnya adalah **GoTo Label**, dimana **Label** merupakan baris yang diberi tanda dan diikuti tanda semicolon(:).

- Contoh **GoTo**:

```
Line10:
  :
  :
  GoTo Line10
```



Saat code mencapai pernyataan **GoTo**, program akan menuju ke baris dengan label **Line10**.

Looping pada Visual Basic

Looping dilakukan dengan format **Do/Loop**. Loop digunakan untuk operasi yang harus diulang-ulang beberapa kali. Loop akan melakukan pengulangan perintah sampai syarat tertentu terpenuhi.

- Contoh **Do While/Loop**:

```
Counter = 1
Do While Counter <= 1000
  Debug.Print Counter
  Counter = Counter + 1
Loop
```

Loop akan mengulang perintah selama (**While**) variable Counter <= 1000.

– Contoh **Do Until/Loop**:

```
Counter = 1
Do Until Counter > 1000
  Debug.Print Counter
  Counter = Counter + 1
Loop
```

Loop akan mengulang perintah sampai (**Until**) variable Counter > 1000.

– Contoh **Do/Loop While**:

```
Sum = 1
Do
  Debug.Print Sum
  Sum = Sum + 3
Loop While Sum <= 50
```

Loop akan mengulang **While** jika Variable Sum <= 50. Minimal satu kali perintah akan dilaksanakan.

– Contoh **Do/Loop Until**:

```
Sum = 1
Do
  Debug.Print Sum
  Sum = Sum + 3
Loop Until Sum > 50
```

Loop akan mengulang **Until** jika variable Sum <= 50. Minimal satu kali perintah akan dilaksanakan.

Visual Basic Counting

Pengulangan dengan jumlah tertentu (Counting) dilaksanakan dengan loop **For/Next**.

```
For I = 1 to 50 Step 2
  A = I * 2
  Debug.Print A
Next I
```

Toolbox Visual Basic

Message Box

Salah satu fungsi terbaik dalam Visual Basic adalah **message box**. Message box menampilkan message (pesan), opsional icon, dan beberapa command button. User melakukan respon dengan meng-click sebuah button.

Bentuk pernyataan dari message box tidak menghasilkan nilai (hanya menampilkan pesan):

```
MsgBox Message, Type, Title
```

Dimana:

Message	Teks pesan yang ditampilkan
Type	Jenis dari message box
Title	Teks yang ditampilkan di title bar message box

Pernyataan **Type** merupakan hasil penjumlahan dari empat komponen button yang ditampilkan, icon yang dipergunakan, button yang menjadi default, dan modalitas dari message box. Bentuk **function** dari message box menghasilkan bilangan integer berdasarkan **Type** yang dipergunakan. Contoh dari penggunaan fungsi ini adalah:

```
Dim Response as Integer  
Response = MsgBox(Message, Type, Title)
```

Nilai	Arti	konstanta Simbol
0	OK button	vbOKOnly
1	OK/Cancel button	vbOKCancel
2	Abort/Retry/Ignore button	vbAbortRetryIgnore
3	Yes/No/Cancel button	vbYesNoCancel
4	Yes/No button	vbYesNo
5	Retry/Cancel button	vbRetryCancel

Komponen kedua dari **Type** menentukan **icon** yang hendak ditampilkan pada message box:

Nilai	Arti	Konstanta Simbol
0	No icon	(None)
16	Critical icon	vbCritical
32	Question mark	vbQuestion
48	Exclamation point	vbExclamation
64	Information icon	vbInformation

Komponen ketiga dari **Type** menentukan button yang mana sebagai **default** (yaitu menekan tombol **Enter** sama dengan meng-click default button):

Nilai	Arti	Konstanta Simbol
0	button pertama default	vbDefaultButton1
256	button kedua default	vbDefaultButton2
512	button ketiga default	vbDefaultButton3

Komponen keempat adalah **Type** menentukan **modality**:

Nilai	Arti	Konstanta Simbol
0	Application modal	vbApplicationModal
4096	System modal	vbSystemModal

Jika message box merupakan **Application Modal**, user harus merespon box ini sebelum bekerja pada aplikasi yang sama. Jika message box bersifat **System Modal**, semua aplikasi lain berhenti sampai user merespon message box. Sangat disarankan untuk menggunakan **Konstanta Simbol** daripada menggunakan **Nilai** (numeric).

Nilai yang dihasilkan oleh function message box ini tergantung dengan button yang di-click:

Nilai	Arti	Konstanta Simbol
1	OK button di-click	vbOK
2	Cancel button di-click	vbCancel
3	Abort button di-click	vbAbort
4	Retry button di-click	vbRetry
5	Ignore button di-click	vbIgnore
6	Yes button di-click	vbYes
7	No button di-click	vbNo

Contoh:

MsgBox "Ini adalah contoh message box", vbOKCancel + vbInformation, "Contoh Message Box"



Object Method

Sebuah method merupakan procedure atau function yang melakukan tindakan pada object. Method selalu dituliskan pada code saat run-time. Format untuk memanggil method adalah:

NamaObject.Method

Object Form

Form adalah dimana user interface digambar. Form merupakan pusat dalam mengembangkan aplikasi dalam Visual Basic.

– Form Properties:

Appearance	Pilihan tampilan 3-D atau flat.
BackColor	Mengatur background color.
BorderStyle	Mengatur form border :fixed atau sizeable.
Caption	Mengatur judul dari form window.
Enabled	Jika True, form bisa bereaksi terhadap mouse/keyboard; False:sebaliknya.
Font	Mengatur font type, style, dan size.
ForeColor	Mengatur warna dari teks atau gambar.
Picture	Menempatkan gambar tipe bitmap di form.
Visible	Jika False, Menyembunyikan form.

– Form Event:

Activate	Event Form_Activate dipanggil jika form menjadi active window.
Click	Event Form_Click dipanggil jika form di-click.
DbClick	Event Form_DbClick dipanggil jika form di-double click.
Load	Event Form_Load terjadi jika form di-load oleh aplikasi.

– Form Method:

Cls	Menghapus semua gambar dan teks dari form tetapi tidak menghilangkan object yang lain.
Print	Mencetak teks pada form.

Contoh:

```
frmExample.Cls ' membersihkan form  
frmExample.Print "This will print on the form"
```

Command Button



Command button dipergunakan untuk memulai, menghambat, atau menghentikan suatu proses.

– Command Button Properties:

- Appearance** Pilihan tampilan 3-D atau flat.
- Cancel** Membolehkan menselect button dengan tombol **Esc** (hanya satu button di dalam form yang sama).
- Caption** Mengatur judul dari command button.
- Default** Membolehkan menselect button dengan tombol **Enter** key (hanya satu button di dalam form yang sama).
- Font** Mengatur font type, style, dan size.

– Command Button Event:

- Click** Event yang dipanggil saat button di-select baik dengan meng-click atau dengan access key.

Label Boxes



Label box merupakan control yang dipakai dimana user tidak bisa secara langsung mengeditnya. Teks di dalam label box dapat diubah pada saat run-time.

– Label Properties:

- Alignment** Alignment judul di dalam border.
- Appearance** Pilihan tampilan 3-D atau flat.
- AutoSize** Jika True, label akan berukuran sesuai dengan ukuran teks yang dipakai; False: ukuran dari label tetap.
- BorderStyle** menentukan tipe dari border.
- Caption** Teks yang ditampilkan di dalam label box.
- Font** Mengatur font type, style, dan size.
- WordWrap** Jika AutoSize = True, WordWrap = True, kemudian teks akan tersusun ke bawah dan label akan melebar secara vertikal sesuai dengan teksnya.

– Label Events:

- Click** Event yang dipanggil saat label di-click.
- DblClick** Event yang dipanggil saat label di-double click.

Text Box



Text box dipergunakan untuk menampilkan informasi. Tampilan teks di text box bisa diedit.

– Text Box Properties:

Appearance	Pilihan tampilan 3-D atau flat.
BorderStyle	Menentukan tipe dari border.
Font	Mengatur font type, style, dan size.
MaxLength	Membatasi jumlah teks yang bisa diketikkan (nilai 0 menunjukkan tidak ada pembatasan).
MultiLine	Menentukan tampilan text box satu baris atau banyak baris.
PasswordChar	Menyembuyikan teks dengan suatu karakter tunggal.
ScrollBars	Menentukan scroll bar yang hendak ditampilkan.
SelLength	Panjang dari teks (saat run-time).
SelStart	Posisi awal teks (saat run-time).
SelText	Teks yang diselect (saat run-time).
Tag	Menyimpan ekspresi string.
Text	Teks yang ditampilkan.

– Text Box Event:

Change	Dipanggil saat property Text berubah.
LostFocus	Dipanggil saat user meninggalkan text box.
KeyPress	Dipanggil saat tombol ditekan

– Text Box Method:

SetFocus	Menempatkan cursor pada text box yang ditentukan.
-----------------	---

Contoh:

```
txtExample.SetFocus ' memindahkan cursor ke text box txtExample
```

Line Tool



Line tool dipergunakan untuk membuat garis lurus sederhana dengan variasi ketebalan dan warna.

– Line Tool Properties:

BorderColor	Menentukan warna garis.
BorderStyle	Menentukan bentuk garis. Garis bisa berupa tranparan, solid, putus-putus, titik-titik, atau kombinasi.
BorderWidth	Menentukan ketebalan garis.

Shape Tool



Shape tool dipakai untuk membuat bentuk lingkaran, oval, bujursangkar, persegi panjang, bujursangkar dan persegi panjang dengan sisi tumpul. Warna dan isian lainnya dapat dipakai untuk object ini.

– Shape Tool Properties:


BackColor	Menentukan warna background dari shape (jika dalam mode FillStyle tidak Solid).
BackStyle	Menentukan apakah background transparan atau tidak.
BorderColor	Menentukan warna garis dari shape.
BorderStyle	Menentukan style dari garis shape. Garis ini bisa berupa transparan, solid, putus-putus, titik-titik.
BorderWidth	Menentukan ketebalan dari garis shape.
FillColor	Menentukan warna dalam dari shape.
FillStyle	Menentukan pola isian dari shape.
Shape	menentukan bentuk shape (bujursangkar, persegi panjang, lingkaran, dsb).

Seperti line tool, event dan method tidak dimiliki oleh object ini.

Horizontal dan Vertical Scroll Bar




Horizontal dan vertical **scroll bars** banyak dipakai dalam aplikasi windows. Kedua tipe scroll bars berupa tiga area yang dapat di-Click, seret, untuk mengubah nilai dari scrollbar. Area itu adalah:




– Scroll Bar Properties:

LargeChange	Nilai pertambahan atau pengurangan dari properties scroll bar Value jika bar area di-Click.
Max	Nilai maksimum dari horizontal scroll bar dapat terjadi antara -32,768 s.d 32,767.
Min	Nilai minimum dari horizontal scroll bar dapat terjadi antara -32,768 s.d 32,767.
SmallChange	Nilai pertambahan atau pengurangan dari properties scroll bar Value jika end arrow di-Click.
Value	Posisi terkini dari scrollbox pada scroll bar.

Properties untuk horizontal scroll bar:



Properties untuk vertical scroll bar:



- Scroll Bar Event:
 - Change** Event ini terjadi setelah posisi scroll box telah berubah.
 - Scroll** Event ini terjadi pada saat scroll box digeser.

Check Box



Check box menyediakan cara untuk membuat pilihan dari sebuah daftar pilihan. Sebagian, semua atau tak satupun pilihan bisa di-select

- Check Box Properties:
 - Caption** Judul dari masing-masing pilihan
 - Font** Mengatur font type, style, dan size.
 - Value** Menunjukkan jika unchecked (0, vbUnchecked), checked (1, vbChecked), atau grayed out (2, vbGrayed).
- Check Box Events:
 - Click** Terpanggil jika sebuah box di-click. property Value secara otomatis diubah oleh Visual Basic.

Option Button



Option buttons menyediakan kemampuan untuk membuat pilihan dalam satu grup dan hanya satu saja yang terpilih (True).

– Option Button Properties:

Caption	Judul dari masing-masing pilihan.
Font	Mengatur font type, style, dan size.
Value	Menunjukkan jika ter-select (True) atau tidak (False). Hanya satu button yang bernilai True.

– Option Button Events:

Click	Terpanggil jika sebuah box di-click. property Value secara otomatis diubah oleh Visual Basic.
--------------	---

Array

Array yang kita bicarakan disini, merupakan yang dipergunakan untuk variable yang mempunyai dimensi yang banyak dengan satu nama saja. Array dideklarasikan sama dengan variable biasa dengan tambahan (**dimensi**). Sebagai contoh kita membuat variable array dengan nama '**Items**', dengan dimensi **9**, pada level procedure:

```
Dim Items(9) as Integer
```

Jika kita tetap ingin variable array tetap mempertahankan nilai yang disimpennya saat procedure dimatikan, kita bisa gunakan keyword **Static**:

```
Static Items(9) as Integer
```

Pada level **form** atau **module**, di **general declarations**, gunakan :

```
Dim Items(9) as Integer
```

Dan pada level module, untuk **global** declaration, gunakan:

```
Global Items(9) as Integer
```

Index dari variable array mulai angka 0 dan berakhir dengan nilai dimensinya. Contoh, array **Items** pada contoh di atas mempunyai sepuluh elemen, mulai dari **Items(0)** to **Items(9)**. variable array dalam penggunaannya sama dengan variable lainnya dengan menambahkan index. Contoh kita hendak memberi nilai pada Items(5) sama dengan 100, maka code yang dipakai:

```
Items(5)=100
```

Control Array

Control array merupakan penggunaan array pada control-control yang mempunyai fungsi sama. Cara melakukannya dengan membuat control-control dengan tipe yang sama. Kemudian diberi nama yang sama dan menambahkan index pada property masing-masing control tersebut.

Frame



Frame dipakai untuk mengelompokkan berbagai object di dalam satu grup/tempat. Cara yang dipakai untuk menggunakan frame, pertama gambarkan frame di dalam form. Kemudian **Copy** object yang hendak dimasukkan ke frame, lalu click frame. **Paste** object, maka semua object yang dicopykan ke dalam frame akan mengikuti perilaku dari frame tersebut.

– Frame Properties:

- Caption** Judul dari frame.
- Font** Mengatur jenis font, style, dan ukuran.

List Box



Sebuah **list box** menampilkan sederet item dimana user dapat men-select satu atau lebih item. Jika jumlah item lebih banyak daripada yang dapat ditampilkan, sebuah scroll bar otomatis akan ditampilkan.

– List Box Properties:

- Appearance** Tampilan 3-D atau flat.
- List** Array dari item di dalam list box.
- ListCount** Jumlah item di dalam list.
- ListIndex** Index dari list yang akan ditampilkan, jika tidak ada otomatis ListIndex=-1.
- MultiSelect** Mengontrol bagaimana item bisa di-select (0-no multiple selection allowed, 1-multiple selection allowed, 2-group selection allowed).
- Selected** Jika ada item yang di-select berarti True, dan sebaliknya berarti False.
- Sorted** True berarti diurutkan berdasarkan ASCII.
- Text** Text dari item yang di-select.

– List Box Event:

- Click** Event dipanggil saat sebuah item di dalam list box di-click.
- DbClick** Event dipanggil saat sebuah item di dalam list box di-double click

– List Box Method:

- AddItem** Menambahkan sebuah item di dalam list.
- Clear** Menghapus semua item di dalam list box.
- RemoveItem** Menghapus sebuah item di dalam list box, berdasarkan index-nya.

Contoh:

```
lstExample.AddItem "This is an added item" lstExample.Clear  
lstExample.RemoveItem 4 ' menghapus lstExample.List(4) dari list box
```

Combo Box



Combo box mirip dengan list box. Perbedaannya adalah sebuah combo box berisi text box pada bagian atas dari list box dan hanya mengijinkan satu item saja untuk di-select.

– Combo Box Properties:

Appearance	Pilihan tampilan 3-D atau flat.
List	Array dari item di bagian list box.
ListCount	Jumlah item di dalam list.
ListIndex	Index dari list yang akan ditampilkan, jika tidak ada otomatis ListIndex=-1.
Sorted	True berarti diurutkan berdasarkan ASCII
Style	Memilih bentuk combo box. Style = 0, Dropdown combo; user bisa mengubah item yang di-select. Style = 1, Simple combo; user bisa mengubah item yang di-select. Style = 2, Dropdown combo; user tidak bisa mengubah item yang di-select.
Text	Text dari item yang di-select.

– Combo Box Events:

Click	Event dipanggil saat sebuah item di dalam list box di-click.
DbClick	Event dipanggil saat sebuah item di dalam list box di-double click

– Combo Box Method:

AddItem	Menambahkan sebuah item di dalam list.
Clear	Menghapus semua item di dalam list box.
RemoveItem	Menghapus sebuah item di dalam list box, berdasarkan index-nya.

Contoh:


```
cboExample.AddItem "This is an added item"  
cboExample.Clear ' clears the combo box  
cboExample.RemoveItem 4 ' menghapus cboExample.List(4) dari list
```

MODUL 1


Latihan 1-1

Aplikasi Stopwatch - Menggambar Control

1. Mulai dengan Start → Programs → Microsoft Visual Studio 6 → Microsoft Visual Basic 6



2. Mulai dengan new project. Dasar pemikiran dari project ini adalah untuk menjalankan timer (pencatat waktu), kemudian menghentikan timer dan menghitung waktu selang yang terjadi. Dari form **New Project** pilih **Standard EXE** and click **Open**.



3. Tempatkan tiga command button dan enam label pada form. Atur letak dan ukuran dari control dan form sehingga tampak seperti di bawah ini:



Latihan 1-2


Aplikasi Stopwatch - Setting Properties

1. Set properties dari form, tiga button, dan enam label:

Form1:		Label2:	
BorderStyle	1-Fixed Single	Caption	End Time
Caption	Stopwatch Application	Label3:	
Name	frmStopWatch	Caption	Elapsed Time
Command1:		Label4:	
Caption	&Start Timing	BorderStyle	1-Fixed Single
Name	cmdStart	Caption	[Blank]
Command2:		Name	lblStart
Caption	&End Timing	Label5:	
Name	cmdEnd	BorderStyle	1-Fixed Single
Command3:		Caption	[Blank]
Caption	E&xit	Name	lblEnd
Name	cmdExit	Label6:	
Label1:		BorderStyle	1-Fixed Single
Caption	Start Time	Caption	[Blank]
		Name	lblElapsed

Pada properties **Caption** dari tiga command button, perhatikan tanda ampersand (&). Tanda ampersand mengawali **Tombol Akses** untuk button. Yaitu, selain melakukan click pada button untuk memicu suatu event terjadi, kita juga bisa menggunakan tombol akses (tidak perlu pakai mouse). tombol akses ditekan bersamaan dengan tombol **Alt**. Sehingga untuk memanggil 'Begin Timing', kita bisa melakukan click pada button atau tekan tombol Alt+B. Perhatikan, pada form, tombol akses tampak pada caption button dalam bentuk garis bawah (underline).

1. Form anda akan tampak seperti di bawah ini:



2. Simpan pekerjaan anda

Latihan 1-3

Aplikasi Stopwatch - Menuliskan Code

Sekarang tinggal menambahkan code pada aplikasi anda. Kita menuliskan code untuk setiap event yang diperlukan. Dalam aplikasi ini ada tiga event yang diperlukan yaitu: click pada setiap command button.

1. Double-click pada form untuk membuka code window, atau pilih 'View Code' dari project window.
2. Click down arrow pada Object box (**General**), maka procedure box akan muncul (**Declarations**). Tuliskan kode di bawah ini:

```
Option Explicit
Dim StartTime As Variant
Dim EndTime As Variant
Dim ElapsedTime As Variant
```

Pernyataan **Option Explicit** memaksa kita untuk mendeklarasikan semua variable. Baris yang lain membuat variable **StartTime**, **EndTime**, dan **ElapsedTime** sebagai variable global di dalam form.

3. Select object **cmdStart** pada Object box. Jika procedure yang muncul bukan procedure Click, pilih **Click** dari procedure box. Ketikkan code berikut ini. Perhatikan bahwa statemen **Sub** dan **End Sub** otomatis disediakan oleh Visual Basic:

```
Sub cmdStart_Click ()
    'membuat dan menampilkan starting time
    StartTime = Now
    lblStart.Caption = Format(StartTime, "hh:mm:ss")
    lblEnd.Caption = ""
    lblElapsed.Caption = ""
End Sub
```

4. Sekarang, code untuk button **cmdEnd**.

```
Sub cmdEnd_Click ()
    'Temukan ending time, menghitung waktu tempuh
    'menampilkan kedua nilai pada label
    EndTime = Now
    ElapsedTime = EndTime - StartTime
    lblEnd.Caption = Format(EndTime, "hh:mm:ss")
    lblElapsed.Caption = Format(ElapsedTime, "hh:mm:ss")
End Sub
```

Disini, jika button **End Timing** di-click, kita membaca waktu kini (**End Time**), menghitung waktu tempuh dan menampilkan kedua nilai pada label.

5. Akhirnya button **cmdExit**.

```
Sub cmdExit_Click ()
    End ' untuk mengakhiri aplikasi/program
End Sub
```


6. Click toolbar **Run** atau tekan <f5> untuk menjalankan program ini.

Latihan 1-4

Calendar/Time Display

Desain program yang menampilkan hari, bulan, dan tahun sekarang. Juga menampilkan waktu saat ini, berubah tiap satu detik (lihat pada control **Timer**). Buat window tersebut tampak seperti halaman kalender. Susun properties dari object-object yang ada untuk tampil dengan menarik.

Form:



Properties:

Form **frmCalendar**:

Caption My Calendar
BorderStyle 1 - Fixed Single

Timer **timDisplay**:

Interval 1000 ' berubah perdetik

Label **lblDay**:

Caption Sunday
FontName Times New Roman
FontBold True
FontSize 24

Label **lblTime**:

Caption 00:00:00 PM
FontName Times New Roman
FontBold True
FontSize 24

Label **lblYear**:

Alignment 2 - Center
Caption 1998
FontName Times New Roman
FontBold True
FontSize 24

Label **lblNumber**:

Alignment 2 - Center
Caption 31
FontName Arial
FontBold True
FontSize 72

Label **lblMonth**:

Alignment 2 - Center
Caption March
FontName Times New Roman
FontBold True
FontSize 24

Code:

General Declarations:

```
Option Explicit
```

timDisplay Timer Event:

```
Private Sub timDisplay_Timer()  
Dim Today As Variant  
    Today = Now  
    lblDay.Caption = Format(Today, "dddd")  
    lblMonth.Caption = Format(Today, "MMMM")  
    lblYear.Caption = Format(Today, "YYYY")  
    lblnumber.Caption = Format(Today, "d")  
    lblTime.Caption = Format(Today, "h:mm:ss ampm")  
End Sub
```

MODUL 2

Contoh 2-1

Tegangan Tarik Baja

1. Mulai dengan **new project**. Tujuan dari project ini adalah membuat aplikasi untuk menentukan nilai tegangan pada suatu batang baja akibat mendapat beban. Rumus matematik untuk mencari tegangan pada penampang batang tarik adalah:

$$\text{Tegangan} = \frac{P_{\max}}{0.85 \cdot F_{\text{profil}}} < \text{Tegangan ijin}$$

dimana:


Tegangan : Tegangan Tarik (kg/cm²)

Tegangan ijin : Tegangan tarik yang diijinkan pada tipe baja tertentu (kg/cm²)

Pmax : Gaya Tarik maksimum (kg)

Fprofil : luas penampang profil (cm²)

2. Tempatkan 4 label, 4 textbox, dan 2 command button pada form. Sehingga tampak seperti di bawah ini:




The image shows a screenshot of a Windows application window titled "Form1". The window contains a grid of controls. On the left side, there are four labels: "Label1", "Label2", "Label3", and "Label4". To the right of each label is a corresponding text box: "Text1", "Text2", "Text3", and "Text4". Below these text boxes are two command buttons: "Command1" and "Command2". The window has a standard Windows title bar with minimize, maximize, and close buttons.

3. Set properties form dan masing-masing object.

Form1:		Text2:	
BorderStyle	1-Fixed Single	Text	[Blank]
Caption	Tegangan Tarik Baja	Name	txtLuasProfil
Name	frmTegTarikBaja	Text3:	
Label1:		Text	[Blank]
Caption	Beban Maksimum (kg)	Name	txtTegIjin
Label2:		Text4:	
Caption	Luas Profil (cm2)	Text	[Blank]
Label3:		Name	txtTeg
Caption	Tegangan Ijin (kg/cm2)	Command1:	
Label4:		Caption	&Hitung
Caption	Tegangan (kg/cm2)	Name	cmdHitung
Text1:		Command2:	
Text	[Blank]	Caption	E&xit
Name	txtBeban	Name	cmdExit

Sekarang form kita tampak seperti ini:



4. Deklarasikan empat variable di dalam **general declarations**.

```
Option Explicit
Dim Beban As Single
Dim LuasProfil As Single
Dim TegIjin As Single
Dim Tegangan As Single
```

5. Tuliskan kode pada event Click **cmdHitung**.

```
Private Sub cmdHitung_Click ()  
    Beban = Val(txtBeban.Text)  
    LuasProfil = Val(txtLuasProfil.Text)  
    TegIjin = Val(txtTegIjin.Text)  
    Tegangan = Beban/(0.85*LuasProfil)  
    txtTeg.Text = Format(Tegangan, "#####0.00")  
End Sub
```

Kode ini membaca tiga nilai input dari textbox, dan menghitung Tegangan yang terjadi pada profil baja.

6. Tuliskan kode berikut ini pada event Click **cmdExit**.

```
Private Sub cmdExit_Click ()  
End  
End Sub
```

7. Coba program, cek apakah program berjalan dengan baik. Simpan program.

Contoh 2-2

Tegangan Tarik Baja - Key Trapping

1. Perhatikan code ASCII yang bisa diterima adalah 48 s.d 57 (angka), 46 (titik desimal), dan 8 (tombol backspace). Untuk tombol backspace, tambahkan konstanta berikut di area **general declarations** :

```
Const vbKeyDecPt = 46
```

2. Tambahkan kode berikut ini pada tiga procedure: **txtBeban_KeyPress**, **txtLuasProfil_KeyPress**, and **txtTegIjin_KeyPress**.

```
Private Sub txtBeban_KeyPress(KeyAscii As Integer)
If (KeyAscii >= vbKey0 And KeyAscii <= vbKey9) Or KeyAscii = vbKeyDecPt
Or KeyAscii = vbKeyBack Then
Exit Sub
Else
KeyAscii = 0
Beep
End If
End Sub
```

```
Private Sub txtLuasProfil_KeyPress(KeyAscii As Integer)
If (KeyAscii >= vbKey0 And KeyAscii <= vbKey9) Or KeyAscii = vbKeyDecPt
Or KeyAscii = vbKeyBack Then
Exit Sub
Else
KeyAscii = 0
Beep
End If
End Sub
```

```
Private Sub txtTegIjin_KeyPress(KeyAscii As Integer)
If (KeyAscii >= vbKey0 And KeyAscii <= vbKey9) Or KeyAscii = vbKeyDecPt
Or KeyAscii = vbKeyBack Then
Exit Sub
Else
KeyAscii = 0
Beep
End If
End Sub
```

3. Run Program dan cek hasil dari key trapping.

Contoh 2-3


Tegangan Tarik Baja - Keputusan

1. Disini , kita mengubah project Tegangan Tarik Baja, dengan menambahkan sebuah command button yang akan mengosongkan semua textbox. Tambahkan properties berikut ini:

Command3:

Caption	&Clear
Name	cmdClear

Bentuk form akan tampak seperti di bawah ini:



2. Code untuk event **Click** pada button **cmdClear**:

```
Private Sub cmdClear_Click ()  
    txtBeban.Text = ""  
    txtLuasProfil.Text = ""  
    txtTegIjin.Text = ""  
    txtTeg.Text = ""  
End Sub
```

3. Code event **KeyPress** untuk object **txtTeg**:

```
Private Sub txtTeg_KeyPress (KeyAscii As Integer)
    `hanya membolehkan angka, tanda titik atau backspace
    If (KeyAscii>=vbKey0 And KeyAscii<=vbKey9) Or KeyAscii= vbKeyDecPt Or
    KeyAscii = vbKeyBack Then
        Exit Sub
    Else
        KeyAscii = 0
        Beep
    End If
End Sub
```

Kita memerlukan code ini karena sekarang kita bisa memasukkan data pada textbox txtTeg.

4. code untuk event **Click** pada button **cmdHitung** adalah:

```
Private Sub cmdHitung_Click()

    Beban = Val(txtBeban.Text)
    LuasProfil = Val(txtLuasProfil.Text)
    TegIjin = Val(txtTegIjin.Text)
    Tegangan = Beban / (0.85 * LuasProfil)
    txtTeg.Text = Format(Tegangan, "#####0.00")

    If txtBeban.Text = "" Then                                `beban kosong
        Beban = Tegangan*0.85*LuasProfil
        txtBeban.Text = Format(Beban, "#####0.00")
    ElseIf txtLuasProfil.Text = "" Then
        LuasProfil=Beban/(0.85*Tegangan)
        txtLuasProfil.Text = Format(LuasProfil, "##0.00")
    ElseIf txtTeg.Text = "" Then
        Tegangan = Beban / (0.85 * LuasProfil)
        txtTeg.Text = Format(Tegangan, "#####0.00")
    End If
End Sub
```

Dalam code ini apabila ada salah satu textbox tidak berisi nilai maka akan bisa diisi saat event Click pada cmdHitung terjadi.

5. Test dan simpan aplikasi anda.

MODUL 3

Latihan 3-1

Menghitung Rerata dan Standar Deviasi

Buat sebuah aplikasi dimana user bisa memasukkan beberapa angka secara berurutan. Saat selesai memasukkan angka, program menghitung nilai rerata data dan standar deviasi. Jika sebanyak N data yang dimasukkan dengan variable angka X_i , maka rumus untuk menentukan nilai rerata data (\bar{X}) adalah:


$$\bar{x} = \left(\sum_{i=1}^N x_i \right) / N$$

dan untuk standar deviasi (s), cari akar dari persamaan ini:

$$s^2 = [N \sum_{i=1}^N x_i^2 - \left(\sum_{i=1}^N x_i \right)^2] / [N(N - 1)]$$

Solusi

Form:



Properties:

Form **frmStats**:

Caption = Rerata Angka dan Standar Deviasi

CommandButton **cmdExit**:

Caption = E&xit

CommandButton **cmdAmbil**:

Caption = &Ambil Angka

CommandButton **cmdHitung**:

Caption = &Hitung

Label **lblRerata**:

Alignment = 2 - Center

BackColor = &H00FFFFFF& (White)

BorderStyle = 1 - Fixed Single

FontName = MS Sans Serif

FontSize = 12

Label **Label4**:

Caption = Rerata

CommandButton **cmdBaru:**

Caption = &Data Baru

TextBox **txtInput:**

FontName = MS Sans Serif

FontSize = 12

Label **lblStdDev:**

Alignment = 2 - Center

BackColor = &H00FFFFFF& (White)

BorderStyle = 1 - Fixed Single

FontName = MS Sans Serif

FontSize = 12

Label **Label6:**

Caption = Standar Deviasi

Label **lblAngka:**

Alignment = 2 - Center

BackColor = &H00FFFFFF& (White)

BorderStyle = 1 - Fixed Single

FontName = MS Sans Serif

FontSize = 12

Label **Label2:**

Caption = Masukkan Angka

Label **Label1:**

Caption = Jumlah Data

Code:

```
Option Explicit
```

```
Dim JumlahData As Integer
```

```
Dim JumlahX As Single
```

```
Dim JumlahX2 As Single
```

```
Const vbKeyMinus = 45
```

```
Const vbKeyDecPt = 46
```

```
Private Sub cmdAmbil_Click()
```

```
Dim Angka As Single
```

```
txtInput.SetFocus
```

```
JumlahData = JumlahData + 1
```

```
lblAngka.Caption = Str(JumlahData)
```

```
Angka = Val(txtInput.Text)
```

```
JumlahX = JumlahX + Angka
```

```
JumlahX2 = JumlahX2 + Angka ^ 2
```

```
txtInput.Text = ""
```

```
End Sub
```

```
Private Sub cmdHitung_Click()
```

```
Dim Rerata As Single
```

```
Dim StdDev As Single
```

```
txtInput.SetFocus
```

```
If JumlahData < 2 Then
```

```
    Beep
```

```
    Exit Sub
```

```
End If
```

```
Rerata = JumlahX / JumlahData
```

```
lblRerata.Caption = Str(Rerata)
```

```
StdDev = Sqr((JumlahData * JumlahX2 - JumlahX ^ 2) / (JumlahData *  
(JumlahData - 1)))
```

```
lblStdDev.Caption = Str(StdDev)
```

```
End Sub
```

```
Private Sub cmdExit_Click()
```

```
End
```

```
End Sub
```

```
Private Sub cmdBaru_Click()  
txtInput.SetFocus  
JumlahData = 0  
lblAngka.Caption = "0"  
txtInput.Text = ""  
lblRerata.Caption = ""  
lblStdDev.Caption = ""  
JumlahX = 0  
JumlahX2 = 0  
End Sub
```

```
Private Sub txtInput_KeyPress(KeyAscii As Integer)  
If (KeyAscii >= vbKey0 And KeyAscii <= vbKey9) Or KeyAscii = vbKeyMinus  
Or KeyAscii = vbKeyDecPt Or KeyAscii = vbKeyBack Then  
Exit Sub  
ElseIf KeyAscii = vbKeyReturn Then  
Call cmdAmbil_Click  
Else  
KeyAscii = 0  
End If  
End Sub
```

MODUL 4


Latihan 4-1

Kalkulator Sederhana

Buat sebuah aplikasi kalkulator sederhana dimana user bisa melakukan perhitungan penjumlahan, perkalian, pengurangan dan pembagian.

Solusi

Form:



Properties:

Form **frmKalkulator:**

BorderStyle=4

Caption = Kalkulator Sederhana

Maxbutton=False

Minbutton=false

CommandButton **cmdExit:**

Caption = E&xit

CommandButton **cmdHitung:**

Caption = &=

CommandButton **cmdClear:**

Caption = &C

‘untuk cmdAngka

CommandButton **cmdAngka:**

Caption = 0 s.d 9

Index=0 s.d 9

CommandButton **cmdOp(0):**

Caption = +

Index=0

CommandButton **cmdOp(1):**

Caption = -

Index=1

CommandButton **cmdOp(2):**

Caption = *

Index=2

CommandButton **cmdOp(3):**

Caption = sqrt

Index=3

CommandButton **cmdMinPlus:**

Caption = +/-

CommandButton **cmdDec:**

Caption = .

Label **lblHasil:**

BackColor = &H00FFFFFF& (White)

BorderStyle = 1 - Fixed Single

Code:

```
Option Explicit
Dim Angka_1 As Single
Dim Angka_2 As Single
Dim Operator As String

Private Sub Cmdangka_Click(Index As Integer)
    `menampilkan angka di label
    lblHasil.Caption = lblHasil.Caption & cmdAngka(Index).Caption
End Sub

Private Sub cmdClear_Click() `menghilangkan semua nilai pada variable
Angka_1 = 0
Angka_2 = 0
Operator = ""
lblHasil.Caption = ""
End Sub

Private Sub cmdDec_Click() `tanda desimal pada bilangan pecahan
lblHasil.Caption = lblHasil.Caption & "."
End Sub

Private Sub cmdPlusMin_Click()
    `memberi tanda minus bila bilangan negatif
    If Left(lblHasil.Caption, 1) <> "-" Then
        lblHasil.Caption = "-" & lblHasil.Caption
    Else
        lblHasil.Caption = Mid(lblHasil.Caption, 2)
    End If
End Sub

Private Sub cmdOp_Click(Index As Integer)
    `menyimpan isi label ke variable ke-1 dan mengosongkan label
    Angka_1 = Val(lblHasil.Caption)
    lblHasil.Caption = ""
    `menyimpan kode operator aritmatik atau langsung perhit. akar `kuadrat
    Select Case Index
        Case 0: Operator = "+"
        Case 1: Operator = "-"
        Case 2: Operator = "*"
        Case 3: lblHasil.Caption = Str(Sqr(Angka_1))
    End Select
End Sub

Private Sub cmdHitung_Click()
    Angka_2 = Val(lblHasil.Caption) `menyimpan isi label ke variable ke-2
    Select Case Operator `melakukan perhitungan pada ke-dua variable
        Case "+": lblHasil.Caption = Angka_1 + Angka_2
        Case "-": lblHasil.Caption = Angka_1 - Angka_2
        Case "*": lblHasil.Caption = Angka_1 * Angka_2
    End Select
End Sub

Private Sub cmdExit_Click()
    Unload Me `keluar dari program
End Sub
```

Latihan 5-1

Konversi Temperature

Mulai dengan new project. Dalam project ini, kita akan melakukan konversi temperatur dari derajat Fahrenheit (dengan scroll bar) ke derajat Celsius.


formula untuk konversi Fahrenheit (F) ke Celsius (C) adalah:

$$C = (F - 32) * 5/9$$

Untuk mengkonversi angka ke bilangan bulat, gunakan Visual Basic **CInt()** function. Untuk mengubah informasi dalam bentuk numeric ke string, bisa menggunakan function **Str()** atau **Format()**.

Solusi:

1. Buat shape, vertical scroll bar, empat label, dan sebuah command button pada form.



2. Set properties dari form dan tiap-tiap object:

Form1:

BorderStyle	1-Fixed Single
Caption	Temperature Conversion
Name	frmTemp

VScroll1:

LargeChange	10
Max	-60
Min	120
Name	vsbTemp
SmallChange	1
Value	32

Shape1:

BackColor	White
BackStyle	1-Opaque
FillColor	Red
FillStyle	7-Diagonal Cross
Shape	4-Rounded Rectangle

Label1:

Alignment	2-Center
Caption	Fahrenheit
FontSize	10
FontStyle	Bold


Label2:	
Alignment	2-Center
AutoSize	True
BackColor	White
BorderStyle	1-Fixed Single
Caption	32
FontSize	14
FontStyle	Bold
Name	lblTempF

Label4:	
Alignment	2-Center
AutoSize	True
BackColor	White
BorderStyle	1-Fixed Single
Caption	0
FontSize	14
FontStyle	Bold
Name	lblTempC

Label3:	
Alignment	2-Center
Caption	Celsius
FontSize	10
FontStyle	Bold

Command1:	
Cancel	True
Caption	E&xit
Name	cmdExit

Perlu diketahui bahwa temperatur diawali dengan 32F dan 0C.



3. Tulis kode berikut ini ke dalam general declarations.

```
Option Explicit
Dim TempF As Integer
Dim TempC As Integer
```

Hal ini dilakukan untuk membuat ke-dua variable bersifat global.

4. Tambahkan code berikut ini padascroll bar **Scroll** event.

```
Private Sub vsbTemp_Scroll()
    TempF = vsbTemp.Value
    lblTempF.Caption = Str(TempF)
    TempC = CInt((TempF - 32) * 5 / 9)
    lblTempC.Caption = Str(TempC)
End Sub
```

Code ini menentukan Value scroll bar saat mengalami scroll, membaca nilai dari derajat Fahrenheit kemudian menghitung derajat Celsius dan menampilkan nilai ke-duanya.

5. Tambahkan code berikut ini pada scroll bar **Change** event.

```
Private Sub vsbTemp_Change()  
    TempF = vsbTemp.Value  
    lblTempF.Caption = Str(TempF)  
    TempC = Cint((TempF - 32) * 5 / 9)  
    lblTempC.Caption = Str(TempC)  
End Sub
```

Code ini sama dengan scroll event.

6. Tambahkan code berikut ini untuk menutup program pada procedure **cmdExit_Click**.


```
Private Sub cmdExit_Click()  
    End  
End Sub
```

7. Run program dan simpan project anda.

Latihan 5-1

Rencana Penerbangan

1. Mulai new project. Pada contoh ini, penumpang memilih kota tujuan, pesawat yang dipakai dan jam terbang untuk transportasi udara.
2. Tempatkan sebuah list box, dua combo box, tiga label box dan dua command button pada form. Form tersebut tampak seperti di bawah ini:



3. Atur form dan properties dari object- object:

Form1:

BorderStyle 1-Fixed Single
Caption Rencana Penerbangan
Name frmTransUdara

List1:

Name IstKota
Sorted True

Combo1:

Name cboPesawat
Style 2-Dropdown List

Label1:

Caption Kota Tujuan

Combo2:

Name cboJamTerbang
Style 1-Simple
Text [Blank]

Label2:

Caption Perusahaan Penerbangan

Label3:

Caption Jam Terbang

Command1:


Caption &Pilih
Name cmdPilih

Command2:

Caption E&xit
Name cmdExit

(ubah ukuran agar cukup untuk menampung 4 s.d 5 item)

Sehingga tampak seperti ini:



4. Tambahkan code berikut ini ke procedure **Form_Load**:

```
Private Sub Form_Load()  
    'menambahkan item kota tujuan pada list  
    lstKota.Clear  
    lstKota.AddItem "Aceh"  
    lstKota.AddItem "Medan"  
    lstKota.AddItem "Riau"  
    lstKota.AddItem "Palembang"  
    lstKota.AddItem "Lampung"  
    lstKota.AddItem "Bandung"  
    lstKota.AddItem "Semarang"  
    lstKota.AddItem "Yogyakarta"  
    lstKota.AddItem "Surabaya"  
    lstKota.AddItem "Solo"  
    lstKota.AddItem "Bali"  
    lstKota.AddItem "Lombok"  
    lstKota.AddItem "Mataram"  
    lstKota.AddItem "Pesawattle"  
    lstKota.ListIndex = 0  
  
    'menambahkan item pesawat pada list  
    cboPesawat.AddItem "Garuda Indonesia Airways"  
    cboPesawat.AddItem "Merpati Nusantara Airlines"  
    cboPesawat.AddItem "Mandala Airlines"  
    cboPesawat.AddItem "Batavia Air"  
    cboPesawat.AddItem "Adam Air"  
    cboPesawat.AddItem "Lion Air"  
  
    cboPesawat.ListIndex = 0  
    'menambahkan jam terbang ke list  
    cboJamTerbang.AddItem "08.00"  
    cboJamTerbang.AddItem "08.30"  
    cboJamTerbang.AddItem "09.00"  
    cboJamTerbang.AddItem "09.30"  
    cboJamTerbang.AddItem "10.00"
```

```

cboJamTerbang.AddItem "10.30"
cboJamTerbang.AddItem "11.00"
cboJamTerbang.AddItem "11.30"
cboJamTerbang.AddItem "12.00"
cboJamTerbang.AddItem "12.30"
End Sub

```

5. Tambahkan code berikut ini ke procedure **cmdPilih_Click**:

```

Private Sub cmdAssign_Click()
Dim Message As String
If cboJamTerbang.Text = "" Then
    MsgBox "Pilih jam terbang dulu...!", vbOKOnly + vbInformation,
    "Peringatan"
Exit Sub
End If
Message = "tujuan: " + lstKota.Text + vbCr
Message = Message + "Perusahaan Penerbangan: " + cboPesawat.Text +
vbCr
Message = Message + "JamTerbang: " + cboJamTerbang.Text + vbCr
MsgBox Message, vbOKOnly + vbInformation, "Pilihan Anda"
End Sub

```

Ketika command button **Pilih** di-click, code ini membuat message box yang berisi kota tujuan (dari list box **lstKota**), perusahaan penerbangan (dari **cboPesawat**), dan jam terbang yang dipilih (dari **cboJamTerbang**).

6. Tambahkan code berikut ini ke event **cmdExit_Click**:

```

Private Sub cmdExit_Click()
End
End Sub

```

7. Jalankan Aplikasi. Simpan project.

Latihan 5-2

Tampilan Data Mahasiswa Baru

Sebuah universitas ingin anda membuat suatu tampilan input untuk memasukkan data mahasiswa baru. Informasi yang harus dimasukkan adalah:


1. Nama, Usia, Alamat Kota, dan Jenis Kelamin (Pria atau Wanita)
2. Fakultas (Fakultas Teknik, Fakultas Kedokteran, Fakultas Sastra, Fakultas Ilmu Politik, Fakultas Hukum, Fakultas Pertanian, Fakultas Keguruan, Fakultas Ekonomi, Fakultas MIPA)
3. Olahraga (Lari, Jalan, Bersepeda, Lain-lain)

Atur screen sehingga hanya Nama dan Usia (gunakan text box) dan, mungkin, Kota (gunakan combo box) perlu diketikkan; input yang lain diatur dengan check boxes dan option button. Ketika informasi sudah lengkap, tampilkan semua informasi tersebut ke dalam message box seperti ini:



Solusi:

Form:



Properties:

Form **frmMahasiswa:**

BorderStyle = 1 - Fixed Single
Caption = Profil Mahasiswa

CommandButton **cmdExit:**

Caption = E&xit

Frame **Frame3:**

Caption = Kota Tinggal
FontName = MS Sans Serif
FontBold = True
FontSize = 9.75
FontItalic = True

OptionButton **optFakultas**

Caption = Fakultas Teknik
Index = 0

OptionButton **optFakultas**

Caption = Fakultas Kedokteran
Index = 1

OptionButton **optFakultas**

Caption = Fakultas Sastra
Index = 2

OptionButton **optFakultas**

Caption = Fakultas Ilmu Politik
Index = 3

OptionButton **optFakultas**

Caption = Fakultas Hukum
Index = 4

OptionButton **optFakultas**

Caption = Fakultas Pertanian
Index = 5

OptionButton **optFakultas**

Caption = Fakultas Keguruan
Index = 6

OptionButton **optFakultas**

Caption = Fakultas Ekonomi
Index = 7

ComboBox **choKota:**

Sorted = True
Style = 1 - Simple Combo

CommandButton **cmdNew:**

Caption = &Profil Baru

CommandButton **cmdShow:**

Caption = &Tampilkan

Frame **Frame4:**

Caption = Fakultas
FontName = MS Sans Serif
FontBold = True
FontSize = 9.75
FontItalic = True

OptionButton **optSex:**

Caption = Wanita
Index = 1

OptionButton **op... Frame4**

Caption = Pria
Index = 0
Value = True

Frame **Frame2:**

Caption = Activities
FontName = MS Sans Serif
FontBold = True
FontSize = 9.75
FontItalic = True

CheckBox **chkAct:**

Caption = Lain-lain
Index = 4

CheckBox **chkAct:**

Caption = Renang
Index = 3

CheckBox **chkAct:**

Caption = sepeda Santai
Index = 2

CheckBox **chkAct:**

Caption = Jalan
Index = 1

OptionButton **optFakultas**
Caption = Fakultas MIPA
Index = 8

Frame **Frame1**:
Caption = Jenis Kelamin
FontBold = True
FontItalic = True

TextBox **txtNama**:
FontName = MS Sans Serif
FontSize = 12

TextBox **txtUsia**:
FontName = MS Sans Serif
FontSize = 12

CheckBox **chkAct**:
Caption = Lari Pagi
Index = 0

Label **Label1**:
Caption = Nama
FontName = MS Sans Serif
FontBold = True
FontSize = 9.75
FontItalic = True

Label **Label2**:
Caption = Usia
FontName = MS Sans Serif
FontBold = True
FontSize = 9.75
FontItalic = True

Code:

General Declarations:

```
Option Explicit  
Dim Fakultas As String
```

cmdExit Click Event:

```
Private Sub cmdExit_Click()  
    End  
End Sub
```

cmdNew Click Event:

```
Private Sub cmdNew_Click()  
    Dim I As Integer  
    txtNama.Text = ""  
    txtUsia.Text = ""  
    For I = 0 To 4  
        chkAct(I).Value = vbUnchecked  
    Next I  
End Sub
```

cmdShow Click Event:

```
Private Sub cmdShow_Click()  
    Dim NoAct As Integer, I As Integer  
    Dim Msg As String, Kelamin As String  
  
    If txtName.Text = "" Then  
        MsgBox "Nama belum diisi.", vbOKOnly + vbCritical, "Belum ada nama"  
        Exit Sub  
    End If
```



```

If txtAge.Text = "" Then
    MsgBox "Masukkan usia!", vbOKOnly + vbCritical, "Usia belum diisi"
    Exit Sub
End If

Msg = "- Nama: " + txtName.Text + "." + vbCr
Msg = Msg + "- Usia: " + Str$(txtAge.Text) + " tahun." + vbCr

If optSex(0).Value = True Then
    Kelamin = "- Jenis Kelamin: Pria."
Else
    Kelamin = "- Jenis Kelamin: Wanita."
End If

Msg = Msg + Kelamin + vbCr
Msg = Msg + "- Tinggal di kota: " + cboKota.Text + "." + vbCr
Msg = Msg + "- Fakultas: " + Fakultas + vbCr

NoAct = 0
For I = 0 To 4
    If chkAct(I).Value = vbChecked Then NoAct = NoAct + 1
Next I
If NoAct > 0 Then
    Msg = Msg + "- Hobbi:" + vbCr
    For I = 0 To 4
        If chkAct(I).Value vbChecked Then Msg=Msg+String$(10,32)+"@ "+
chkAct(I).Caption + vbCr
    Next I
Else
    Msg = Msg + vbCr
End If
MsgBox Msg, vbOKOnly + vbInformation, "Profil Mahasiswa"
End Sub

```

Form_Load Event:

```

Private Sub Form_Load()
    'Load combo box with potential city names
    cboKota.AddItem "Aceh"
    cboKota.AddItem "Medan"
    cboKota.AddItem "Riau"
    cboKota.AddItem "Palembang"
    cboKota.AddItem "Lampung"
    cboKota.AddItem "Bandung"
    cboKota.AddItem "Semarang"
    cboKota.AddItem "Yogyakarta"
    cboKota.AddItem "Surabaya"
    cboKota.AddItem "Solo"
    cboKota.AddItem "Bali"
    cboKota.AddItem "Lombok"
    cboKota.AddItem "Mataram"
    cboKota.ListIndex = 0
    Fakultas = "Fakultas Teknik"
End Sub

```

optFakultas Click Event:

```
Private Sub optFakultas_Click(Index As Integer)
Select Case Index
Case 0
    Fakultas = "Fakultas Teknik"
Case 1
    Fakultas = "Fakultas Kedokteran"
Case 2
    Fakultas = "Fakultas Sastra"
Case 3
    Fakultas = "Fakultas Ilmu Politik"
Case 4
    Fakultas = "Fakultas Hukum"
Case 5
    Fakultas = "Fakultas Pertanian"
Case 6
    Fakultas = "Fakultas Keguruan"
Case 7
    Fakultas = "Fakultas Ekonomi"
Case Else
    Fakultas = "Fakultas MIPA"
End Select
End Sub
```


txtUsia KeyPress Event:

```
Private Sub txtAge_KeyPress(KeyAscii As Integer)
If (KeyAscii>=vbKey0 And KeyAscii<=vbKey9) Or KeyAscii=vbKeyBack Then
    Exit Sub
Else
    KeyAscii = 0
End If
End Sub
```

Latihan 6-1


Diagram Momen

Mulai dengan new project. Dalam project ini, kita akan menghitung besar momen yang terjadi pada suatu gelagar jembatan yang mengalami beban terpusat.



Solusi:

1. Buat empat shape, picture, horizontal scroll bar, tujuh label, dua textbox dan sebuah command button pada form.



2. Set properties form dan tiap-tiap object:

Form1:

BorderStyle 4'Fixed ToolWindow
Caption "Simulasi Momen Maksimum"
MaxButton 0 'False
MinButton 0 'False
Name frmSimulasiMomenMaks

Label1:

AutoSize 1'True
Caption "P="
Name Label lblP

Label2:

Name lblBtgJembatan
Caption "Bentang Jembatan (m)"

Label3:

Name lblBeban
Caption "Beban (kg)"

Label4:

Name lblPBeban
Caption "Posisi Beban (m)"

Label5:

Name lblPosBeban
BackColor &H00FFFFFF&
BorderStyle 1'Fixed Single

Label6:

Name lblBacaMomen
Caption "Momen Terbaca"

Label7:

Name lblMomen
BackColor &H00FFFFFF&
BorderStyle 1'Fixed Single

Shape1:

BorderColor &H80000001&
BorderWidth 2
FillColor &H00808000&
FillStyle 7'Diagonal Cross
Height 375
Left 240
Name Shape shpBentang
Width 5895

Shape2:

BackColor &H80000001&
BorderColor &H000000FF&
BorderWidth 2
FillColor &H008080FF&
FillStyle 0'Solid
Height 615
Left 240
Name Shape shpBeban
Top 240
Width 255

Shape 3:

FillColor &H00808000&
FillStyle 0'Solid
Height 375
Left 120
Name Shape3
Shape 5'Rounded Square
Top 1200
Width 495

Shape4:

FillColor &H00808000&
FillStyle 0'Solid
Height 375
Left 5760
Name Shape1
Shape 2'Oval
Top 1200
Width 375

CommandButton1:

Name cmdExit
Caption "Exit"

PictureBox1:

Name pcMomen
Appearance 0'Flat
BackColor &H80000005&
BorderStyle 0'None
FillColor &H00FFFFFF&
FillStyle 0'Solid
ForeColor &H80000008&
Height 1400
Left 240
ScaleMode 0'User

TextBox1:


Name txtBeban

TextBox2:

Name txtBtgJembatan

HscrollBar1:

Name hscJarak
 Height 375
 Left 240
 Max 1000
 Top 2640
 Width 5895



3. Tulis kode berikut ini ke dalam general declarations.

```
Option Explicit
Const vbKeyDecPt = 46
Dim Beban As Single
Dim btgJembatan As Single
Dim posBeban As Single
Dim GayaMomen As Single
Dim GayaMomenMaks As Single
Dim tgMomen As Single
```

4. Tambahkan code berikut ini pada scroll bar **Scroll** event.

```
Private Sub hscJarak_Change()  
    'jika nilai yang dimasukkan<=0 tidak akan terjadi apa-apa  
    If (Val(txtBtgJembatan) <= 0 Or Val(txtBeban.Text) = 0) Then  
        hscJarak.Value = 0  
        Exit Sub  
    End If  
  
    'menyesuaikan posisi label beban dengan scroll bar  
    hscJarak.Max = Val(txtBtgJembatan.Text)  
    lblP.Left = shpBentang.Left + (hscJarak.Value / hscJarak.Max) *  
        (shpBentang.Width - shpBeban.Width)  
  
    'menyesuaikan posisi gambar beban dengan scroll bar  
    shpBeban.Left = shpBentang.Left + (hscJarak.Value / hscJarak.Max) *  
        (shpBentang.Width - shpBeban.Width)  
  
    'menampilkan posisi terkini dari beban  
    lblPosBeban.Caption = hscJarak.Value  
  
    'perhitungan momen yang terjadi pada posisi beban  
    Beban = Val(txtBeban.Text)  
    btgJembatan = Val(txtBtgJembatan.Text)  
    posBeban = Val(lblPosBeban.Caption)  
    GayaMomen = (Beban * (btgJembatan - posBeban) / btgJembatan) *  
        posBeban  
    'momen maksimum yang terjadi pada bentang jembatan  
    GayaMomenMaks = Beban * btgJembatan / 4  
    'menampilkan besar momen pada titik beban  
    lblMomen.Caption = GayaMomen  
    'penggambaran bidang momen pada picture box  
    tgMomen = pcMomen.Height * (GayaMomen / GayaMomenMaks)  
    pcMomen.Cls  
    pcMomen.Line (0, 0)-(shpBeban.Left, tgMomen)  
    pcMomen.Line (shpBeban.Left, tgMomen)-(pcMomen.Width, 0)  
End Sub
```

Code ini menentukan posisi beban yang diatur dengan scroll bar. Juga menghitung momen yang terjadi serta menghitung momen maksimum. Hasil perhitungan ditampilkan dalam bentuk grafis yaitu bidang momen.

5. Tambahkan kode berikut pada event **KeyPress** pada textbox **txtBeban**, untuk mengantisipasi masukan data yang salah(bukan numeric).

```
Private Sub txtBeban_KeyPress(KeyAscii As Integer)  
    If (KeyAscii >= vbKey0 And KeyAscii <= vbKey9) Or KeyAscii =  
        vbKeyDecPt Or KeyAscii = vbKeyBack Then  
        Exit Sub  
    Else  
        KeyAscii = 0  
        Beep  
    End If  
End Sub
```

6. Tambahkan kode berikut pada event **KeyPress** pada textbox **txtBtgJembatan**, untuk mengantisipasi masukan data yang salah(bukan numeric).

```
Private Sub txtBtgJembatan_KeyPress(KeyAscii As Integer)
If (KeyAscii >= vbKey0 And KeyAscii <= vbKey9) Or KeyAscii = vbKeyBack
Then
Exit Sub
Else
KeyAscii = 0
Beep
End If
End Sub
```

7. Tambahkan kode berikut pada event **LostFocus** pada textbox **txtBeban**, untuk untuk mengupdate informasi label beban (P).

```
Private Sub txtBeban_LostFocus()
lblP.Caption = "P= " & txtBeban & " kg"
End Sub
```

8. Tambahkan code berikut ini untuk menutup program pada procedure **cmdExit_Click**.


```
Private Sub cmdExit_Click()
End
End Sub
```

9. Run program dan simpan project anda.

Latihan 7-1

Rencana Penerbangan

1. Mulai new project. Pada contoh ini, penumpang memilih kota tujuan, pesawat yang dipakai dan jam terbang untuk transportasi udara.
2. Tempatkan sebuah list box, dua combo box, tiga label box dan dua command button pada form. Form tersebut tampak seperti di bawah ini:



3. Atur form dan properties dari object- object:

Form1:

BorderStyle 1-Fixed Single
Caption Rencana Penerbangan
Name frmTransUdara

List1:

Name IstKota
Sorted True

Combo1:

Name cboPesawat
Style 2-Dropdown List

Label1:

Caption Kota Tujuan

Combo2:

Name cboJamTerbang
Style 1-Simple
Text [Blank]

Label2:

Caption Perusahaan Penerbangan

Label3:

Caption Jam Terbang

Command1:

Caption &Pilih
Name cmdPilih

Command2:

Caption E&xit
Name cmdExit

(ubah ukuran agar cukup untuk menampung 4 s.d 5 item)

Sehingga tampak seperti ini:

The screenshot shows a Windows application window titled "Flight Planner". The window contains three main sections: "Kota Tujuan" (Destination City) with a list box labeled "lstKota"; "Perusahaan Penerbangan" (Airline) with a dropdown menu labeled "cboPesawat"; and "Jam Terbang" (Flight Time) with a list box. At the bottom right, there are two buttons: "Pilih" (Select) and "Exit".

4. Tambahkan code berikut ini ke procedure **Form_Load**:

```
Private Sub Form_Load()  
    'menambahkan item kota tujuan pada list  
    lstKota.Clear  
    lstKota.AddItem "Aceh"  
    lstKota.AddItem "Medan"  
    lstKota.AddItem "Riau"  
    lstKota.AddItem "Palembang"  
    lstKota.AddItem "Lampung"  
    lstKota.AddItem "Bandung"  
    lstKota.AddItem "Semarang"  
    lstKota.AddItem "Yogyakarta"  
    lstKota.AddItem "Surabaya"  
    lstKota.AddItem "Solo"  
    lstKota.AddItem "Bali"  
    lstKota.AddItem "Lombok"  
    lstKota.AddItem "Mataram"  
    lstKota.AddItem "Pesawattle"  
    lstKota.ListIndex = 0  
  
    'menambahkan item pesawat pada list  
    cboPesawat.AddItem "Garuda Indonesia Airways"  
    cboPesawat.AddItem "Merpati Nusantara Airlines"  
    cboPesawat.AddItem "Mandala Airlines"  
    cboPesawat.AddItem "Batavia Air"  
    cboPesawat.AddItem "Adam Air"  
    cboPesawat.AddItem "Lion Air"  
  
    cboPesawat.ListIndex = 0  
    'menambahkan jam terbang ke list  
    cboJamTerbang.AddItem "08.00"  
    cboJamTerbang.AddItem "08.30"  
    cboJamTerbang.AddItem "09.00"  
    cboJamTerbang.AddItem "09.30"  
    cboJamTerbang.AddItem "10.00"
```

```

cboJamTerbang.AddItem "10.30"
cboJamTerbang.AddItem "11.00"
cboJamTerbang.AddItem "11.30"
cboJamTerbang.AddItem "12.00"
cboJamTerbang.AddItem "12.30"
End Sub

```

5. Tambahkan code berikut ini ke procedure **cmdPilih_Click**:

```

Private Sub cmdAssign_Click()
Dim Message As String
If cboJamTerbang.Text = "" Then
    MsgBox "Pilih jam terbang dulu...!", vbOKOnly + vbInformation,
    "Peringatan"
Exit Sub
End If
Message = "tujuan: " + lstKota.Text + vbCr
Message = Message + "Perusahaan Penerbangan: " + cboPesawat.Text +
vbCr
Message = Message + "JamTerbang: " + cboJamTerbang.Text + vbCr
MsgBox Message, vbOKOnly + vbInformation, "Pilihan Anda"
End Sub

```

Ketika command button **Pilih** di-click, code ini membuat message box yang berisi kota tujuan (dari list box **lstKota**), perusahaan penerbangan (dari **cboPesawat**), dan jam terbang yang dipilih (dari **cboJamTerbang**).

6. Tambahkan code berikut ini ke event **cmdExit_Click**:

```

Private Sub cmdExit_Click()
End
End Sub

```

7. Jalankan Aplikasi. Simpan project.

Latihan 7-2

Tampilan Data Mahasiswa Baru

Sebuah universitas ingin anda membuat suatu tampilan input untuk memasukkan data mahasiswa baru. Informasi yang harus dimasukkan adalah:


1. Nama, Usia, Alamat Kota, dan Jenis Kelamin (Pria atau Wanita)
2. Fakultas (Fakultas Teknik, Fakultas Kedokteran, Fakultas Sastra, Fakultas Ilmu Politik, Fakultas Hukum, Fakultas Pertanian, Fakultas Keguruan, Fakultas Ekonomi, Fakultas MIPA)
3. Olahraga (Lari, Jalan, Bersepeda, Lain-lain)

Atur screen sehingga hanya Nama dan Usia (gunakan text box) dan, mungkin, Kota (gunakan combo box) perlu diketikkan; input yang lain diatur dengan check boxes dan option button. Ketika informasi sudah lengkap, tampilkan semua informasi tersebut ke dalam message box seperti ini:



Solusi:

Form:



Properties:

Form **frmMahasiswa:**

BorderStyle = 1 - Fixed Single
Caption = Profil Mahasiswa

CommandButton **cmdExit:**

Caption = E&xit

Frame **Frame3:**

Caption = Kota Tinggal
FontName = MS Sans Serif
FontBold = True
FontSize = 9.75
FontItalic = True

OptionButton **optFakultas**

Caption = Fakultas Teknik
Index = 0

OptionButton **optFakultas**

Caption = Fakultas Kedokteran
Index = 1

OptionButton **optFakultas**

Caption = Fakultas Sastra
Index = 2

OptionButton **optFakultas**

Caption = Fakultas Ilmu Politik
Index = 3

OptionButton **optFakultas**

Caption = Fakultas Hukum
Index = 4

OptionButton **optFakultas**

Caption = Fakultas Pertanian
Index = 5

OptionButton **optFakultas**

Caption = Fakultas Keguruan
Index = 6

OptionButton **optFakultas**

Caption = Fakultas Ekonomi
Index = 7

ComboBox **choKota:**

Sorted = True
Style = 1 - Simple Combo

CommandButton **cmdNew:**

Caption = &Profil Baru

CommandButton **cmdShow:**

Caption = &Tampilkan

Frame **Frame4:**

Caption = Fakultas
FontName = MS Sans Serif
FontBold = True
FontSize = 9.75
FontItalic = True

OptionButton **optSex:**

Caption = Wanita
Index = 1

OptionButton **op... Frame4**

Caption = Pria
Index = 0
Value = True

Frame **Frame2:**

Caption = Activities
FontName = MS Sans Serif
FontBold = True
FontSize = 9.75
FontItalic = True

CheckBox **chkAct:**

Caption = Lain-lain
Index = 4

CheckBox **chkAct:**

Caption = Renang
Index = 3

CheckBox **chkAct:**

Caption = sepeda Santai
Index = 2

CheckBox **chkAct:**

Caption = Jalan
Index = 1

OptionButton **optFakultas**
Caption = Fakultas MIPA
Index = 8

Frame **Frame1**:
Caption = Jenis Kelamin
FontBold = True
FontItalic = True

TextBox **txtNama**:
FontName = MS Sans Serif
FontSize = 12

TextBox **txtUsia**:
FontName = MS Sans Serif
FontSize = 12

CheckBox **chkAct**:
Caption = Lari Pagi
Index = 0

Label **Label1**:
Caption = Nama
FontName = MS Sans Serif
FontBold = True
FontSize = 9.75
FontItalic = True

Label **Label2**:
Caption = Usia
FontName = MS Sans Serif
FontBold = True
FontSize = 9.75
FontItalic = True

Code:

General Declarations:

```
Option Explicit  
Dim Fakultas As String
```

cmdExit Click Event:

```
Private Sub cmdExit_Click()  
    End  
End Sub
```

cmdNew Click Event:

```
Private Sub cmdNew_Click()  
    Dim I As Integer  
    txtNama.Text = ""  
    txtUsia.Text = ""  
    For I = 0 To 4  
        chkAct(I).Value = vbUnchecked  
    Next I  
End Sub
```

cmdShow Click Event:

```
Private Sub cmdShow_Click()  
    Dim NoAct As Integer, I As Integer  
    Dim Msg As String, Kelamin As String  
  
    If txtName.Text = "" Then  
        MsgBox "Nama belum diisi.", vbOKOnly + vbCritical, "Belum ada nama"  
        Exit Sub  
    End If
```

```

If txtAge.Text = "" Then
    MsgBox "Masukkan usia!", vbOKOnly + vbCritical, "Usia belum diisi"
    Exit Sub
End If

Msg = "- Nama: " + txtName.Text + "." + vbCr
Msg = Msg + "- Usia: " + Str$(txtAge.Text) + " tahun." + vbCr

If optSex(0).Value = True Then
    Kelamin = "- Jenis Kelamin: Pria."
Else
    Kelamin = "- Jenis Kelamin: Wanita."
End If

Msg = Msg + Kelamin + vbCr
Msg = Msg + "- Tinggal di kota: " + cboKota.Text + "." + vbCr
Msg = Msg + "- Fakultas: " + Fakultas + vbCr

NoAct = 0
For I = 0 To 4
    If chkAct(I).Value = vbChecked Then NoAct = NoAct + 1
Next I
If NoAct > 0 Then
    Msg = Msg + "- Hobbi:" + vbCr
    For I = 0 To 4
        If chkAct(I).Value vbChecked Then Msg=Msg+String$(10,32)+"@ "+
chkAct(I).Caption + vbCr
    Next I
Else
    Msg = Msg + vbCr
End If
MsgBox Msg, vbOKOnly + vbInformation, "Profil Mahasiswa"
End Sub

```

Form_Load Event:

```

Private Sub Form_Load()
    'Load combo box with potential city names
    cboKota.AddItem "Aceh"
    cboKota.AddItem "Medan"
    cboKota.AddItem "Riau"
    cboKota.AddItem "Palembang"
    cboKota.AddItem "Lampung"
    cboKota.AddItem "Bandung"
    cboKota.AddItem "Semarang"
    cboKota.AddItem "Yogyakarta"
    cboKota.AddItem "Surabaya"
    cboKota.AddItem "Solo"
    cboKota.AddItem "Bali"
    cboKota.AddItem "Lombok"
    cboKota.AddItem "Mataram"
    cboKota.ListIndex = 0
    Fakultas = "Fakultas Teknik"
End Sub

```

optFakultas Click Event:

```
Private Sub optFakultas_Click(Index As Integer)
Select Case Index
Case 0
    Fakultas = "Fakultas Teknik"
Case 1
    Fakultas = "Fakultas Kedokteran"
Case 2
    Fakultas = "Fakultas Sastra"
Case 3
    Fakultas = "Fakultas Ilmu Politik"
Case 4
    Fakultas = "Fakultas Hukum"
Case 5
    Fakultas = "Fakultas Pertanian"
Case 6
    Fakultas = "Fakultas Keguruan"
Case 7
    Fakultas = "Fakultas Ekonomi"
Case Else
    Fakultas = "Fakultas MIPA"
End Select
End Sub
```

txtUsia KeyPress Event:

```
Private Sub txtAge_KeyPress(KeyAscii As Integer)
If (KeyAscii>=vbKey0 And KeyAscii<=vbKey9) Or KeyAscii=vbKeyBack Then
    Exit Sub
Else
    KeyAscii = 0
End If
End Sub
```